
Boîte de sélection de chemins et ou de fichier(s)

Les fonctions OpenFile et SaveFile Permettent d'ouvrir une boîte de dialogue pour récupérer
le chemin complet d'un ou plusieurs fichiers à ouvrir sélectionnés par l'opérateur.

Sommaire
•  Avec Access 2003
•  Avec Access 97

 Access 2003 | Chapitre haut de page |
 Access 97 | Chapitre bas de page |

Avec Access 2003

Version utilisée dans ma Médiathèque.
Trouvé sur le site de Raymond.

Appel de la fonction

 Variable_string = OpenFile([Répertoire_initial] [, Type_Sélection] [, ] [, ] [, ] [, )

 Variable_string = SaveFile([Répertoire_initial] [, ])

Répertoire_Initial indique le nom du répertoire qui sera affiché dans la boite de dialogue.
Type_Sélection indique le type de sélection des fichiers, mono sélection (1 seul fichier) ou multi sélection (plusieurs
fichiers). La boîte SaveFile sera toujours en mone-sélection.
Type de fenêtre indique par True ou False si la boîte de sélection sera une fenêtre modale ou non. La boîte
SaveFile sera toujours affichée en mode modal.
Type de fichier indique le type de fichiers à pré-sélectionner à l'ouverture de la boîte.
La valeur sera sélectionnée entre : AllFiles, Image, MSOffice, MultiMedia et Sound.
KbBufferLength, valeur par défaut = 4, indique la taille maximum en Ko des données retournées par la fonction.
Valeur de 4 à 63 Ko. En mono sélection, la taille sera fixe à 254 caractères.
WithPath indique par True ou False (True par défaut) si le chemin du ou des fichiers doit être indiqué dans la
donnée retournée. Si vous indiquez False, les noms de fichiers seulement sont retournés par la fonction.

Valeurs retournées par l'api

l 'api retournera le nom du fichier ou le chemin de tous les fichiers suivi des noms de fichiers, chaque élément étant

séparé par ;.
– En mono-sélection, si le nombre de caractères retournés est > 254, la valeur retournée sera "" .
– En multi-sélection, si le nombre de caractères retournés est > 4094, la valeur retournée sera "" .

l 'api met à votre disposition deux valeurs, pour la fonction OpenFile :

La variable public nOpenFile qui contient le nombre de fichiers sélectionnés. Si le bouton Annuler a été cliqué la
valeur retourné sera 0.
 Le tableau public tOpenFile de base 0, contenant le chemin et les noms des fichiers si le nombre de fichiers est
supérieur à 1.
l 'élément tOpenFile(0) contient le chemin complet de tous les fichiers sélectionnés.
les éléments tOpenFile(1) à tOpenFile(n) contiennent les noms de fichiers (uniquement les noms).
 pour travailler avec le tableau tOpenFile vous pouvez utiliser les fonctions LBound et UBound.

Exemple d 'appel de fonction

Code à insérer dans un module standard

 Private Sub Commande0_Click()

     Me.lememo = OpenFile(CurrentProject.Path, Multi_Sélection, True, MSOffice, 12, True)

     MsgBox "il y a " & nOpenFile & " fichiers sélectionnés"

End Sub

La taille maximum des valeurs retournées sera de 12 Ko et le chemin sera indiqué dans la valeur retournée.
 Le contrôle Lememo contiendra le chemin complet du fichier à ouvrir en mono-sélection ou le chemin de tous les
fichiers suivi de la liste des fichiers sélectionnés en multi-sélection.
 En multi-sélection tous les noms seront séparés par un ; .
 La fenêtre , ici, sera ouverte en modal. Aucune autre fenêtre ne pourra être activée tant que la fenêtre openfile ne
sera pas fermée.

 Private Sub Commande0_Click()

 Me.lememo = OpenFile(CurrentProject.Path, Multi_Sélection, True, MSOffice, 12, False)

 MsgBox "il y a " & nOpenFile & " fichiers sélectionnés"

End Sub

La taille maximum des valeurs retournées sera de 12 Ko et le chemin ne sera pas indiqué dans la valeur retournée.
Le contrôle Lememo contiendra seulement le nom du fichier à ouvrir en mono-sélection ou seulement les noms de
tous les fichiers en multi-sélection.
En multi-sélection tous les noms seront séparés par un ; .
 La fenêtre , ici, sera ouverte en modal. Aucune autre fenêtre ne pourra être activée tant que la fenêtre openfile ne
sera pas fermée.

Fonction à insérer dans un module standard :

 Option Compare Database

Option Explicit

Public Enum Selection

    Mono_Sélection = 1

    Multi_Sélection = 2

End Enum

Public Enum acOpenFileType

    allfiles

    Image

    MSOffice

    MultiMedia

    Sound

End Enum

Public Const OFN_AllowMultiSelect = &H200

Public Const OFN_CreatePrompt = &H2000

Public Const OFN_EnableHook = &H20

Public Const OFN_EnableTemplate = &H40

Public Const OFN_EnableTemplateHandle = &H80

Public Const OFN_EXPLORER = &H80000

Public Const OFN_ExtensionDifferent = &H400

Public Const OFN_FileMustExist = &H1000

Public Const OFN_HideReadOnly = &H4

Public Const OFN_LongNames = &H200000

Public Const OFN_NoChangeDir = &H8

Public Const OFN_NoDeReferenceLinks = &H100000

Public Const OFN_NoLongNames = &H40000

Public Const OFN_NoNetWorkButton = &H20000

Public Const OFN_NoReadOnlyReturn = &H8000

Public Const OFN_NoTestFileCreate = &H10000

Public Const OFN_NoValiDate = &H100

Public Const OFN_OverWritePrompt = &H2

Public Const OFN_PathMustExist = &H800

Public Const OFN_ReadOnly = &H1

Public Const OFN_ShareAware = &H4000

Public Const OFN_ShareFallThrough = 2

Public Const OFN_ShareNoWarn = 1

Public Const OFN_ShareWarn = 0

Public Const OFN_ShowHelp = &H10

Public Dialogue As OpenFileName

Public strFiltre As String

Public strFile As String

Public strNomFile As String

Public RetVal As Long

Public tOpenFile As Variant

Public nOpenFile As Long

Public FileType As acOpenFileType

Public Type OpenFileName

    lStructSize As Long

    hwndOwner As Long

    Instance As Long

    lpstrFilter As String

    lpstrCustomFilter As String

    nMaxCustomFilter As Long

    nFilterIndex As Long

    lpstrFile As String

    nMaxFile As Long

    lpstrFileTitle As String

    nMaxFileTitle As Long

    lpInitialFolder As String

    lpstrTitle As String

    Flags As Long

    nFileOffset As Integer

    nFileExtension As Integer

    lpstrDefExt As String

    lCustData As Long

    lpfnHook As Long

    lpTemplateName As String

End Type

Public Declare Function GetOpenFileName Lib "comdlg32.dll" _

    Alias "GetOpenFileNameA" (pOpenfilename As OpenFileName) As Long

Public Declare Function GetSaveFileName Lib "comdlg32.dll" _

    Alias "GetSaveFileNameA" (pOpenfilename As OpenFileName) As Long

Private Sub ExtendChoose(FileType As acOpenFileType)

    Select Case FileType

        Case allfiles

            strFiltre = "Tous les fichiers" & Chr$(0) & "*.*"

        Case Image

            strFiltre = _

                "Fichiers Jpg" & Chr$(0) & "*.jpg;*.jpeg" & Chr$(0) & _

                "Fichiers Bmp" & Chr$(0) & "*.bmp" & Chr$(0) & _

                "Fichiers Wmf" & Chr$(0) & "*.wmf" & Chr$(0) & _

                "Fichiers Tif" & Chr$(0) & "*.tif" & Chr$(0) & _

                "Tous les fichiers" & Chr$(0) & "*.*"

        Case MSOffice

            strFiltre = _

                "Fichiers Access" & Chr$(0) & "*.mdb;*.mde" & Chr$(0) & _

                "Fichiers Word" & Chr$(0) & "*.doc;*txt" & Chr$(0) & _

                "Fichiers Excel" & Chr$(0) & "*.xls" & Chr$(0) & _

                "Tous les fichiers" & Chr$(0) & "*.*"

        Case MultiMedia

            strFiltre = _

                "Fichiers Mpg" & Chr$(0) & "*.mpg;*.mpeg" & Chr$(0) & _

                "Fichiers Avi" & Chr$(0) & "*.avi" & Chr$(0) & _

                "Fichiers Wmv" & Chr$(0) & "*.wmv" & Chr$(0) & _

                "Fichiers Asx" & Chr$(0) & "*.asx" & Chr$(0) & _

                "Tous les fichiers" & Chr$(0) & "*.*"

        Case Sound

            strFiltre = _

                "Fichiers Mp3" & Chr$(0) & "*.mp3;*.mpc" & Chr$(0) & _

                "Fichiers Wav" & Chr$(0) & "*.wav" & Chr$(0) & _

                "Tous les fichiers" & Chr$(0) & "*.*"

    End Select

End Sub

Public Function OpenFile(Optional InitialFolder As String = "", _

    Optional MultiSelect As Selection = Mono_Sélection, _

    Optional ModalWindow As Boolean = True, _

    Optional FileType As acOpenFileType = MSOffice, _

    Optional KbBufferLength As Long = 4, _

    Optional WithPath As Boolean = True _

    ) As String

    OpenFile = ""

    If InitialFolder = "" Then InitialFolder = CurrentProject.Path

    If KbBufferLength < 4 Or KbBufferLength > 63 Then

        KbBufferLength = 4

    End If

    ExtendChoose FileType

    With Dialogue

        If ModalWindow Then

            .hwndOwner = Access.hWndAccessApp

        Else

            .hwndOwner = 0

        End If

        .lStructSize = Len(Dialogue)

        .lpstrFilter = strFiltre

        .lpInitialFolder = InitialFolder

        .lpstrTitle = "Recherche d'un fichier"

        If MultiSelect = Mono_Sélection Then

            .lpstrFile = Space(254)

            .nMaxFile = 255

            .lpstrFileTitle = Space(254)

            .nMaxFileTitle = 255

            .Flags = 6148

        Else

            .lpstrFile = Space((KbBufferLength * 4096) - 2)

            .nMaxFile = ((KbBufferLength * 4096) - 1)

            .lpstrFileTitle = Space((KbBufferLength * 4096) - 2)

            .nMaxFileTitle = ((KbBufferLength * 4096) - 1)

            .Flags = 6148 Or OFN_FileMustExist _

                Or OFN_EXPLORER _

                Or OFN_AllowMultiSelect

        End If

    End With

    RetVal = GetOpenFileName(Dialogue)

    If RetVal = 0 Then

        OpenFile = ""

        nOpenFile = 0

        Exit Function

    End If

    OpenFile = Trim(Dialogue.lpstrFile)

    OpenFile = Left(OpenFile, Len(OpenFile) - 1)

    OpenFile = Replace(OpenFile, Chr(0), ";")

    tOpenFile = Split(OpenFile, ";")

    If UBound(tOpenFile) = 0 Then

        nOpenFile = 1

    ElseIf UBound(tOpenFile) = 1 Then

        OpenFile = Replace(OpenFile, ";", "")

        nOpenFile = 1

    Else

        nOpenFile = UBound(tOpenFile) - 1

    End If

    If Not WithPath Then

        If InStr(1, OpenFile, ";") > 0 Then

            OpenFile = Mid(OpenFile, InStr(1, OpenFile, ";") + 1)

        Else

            OpenFile = Mid(OpenFile, InStrRev(OpenFile, "\") + 1)

        End If

    End If

End Function

Public Function SaveFile(Optional InitialFolder As String = "", _

    Optional FileType As acOpenFileType = allfiles _

    ) As String

    SaveFile = ""

    If InitialFolder = "" Then InitialFolder = CurrentProject.Path

    ExtendChoose FileType

    With Dialogue

        .hwndOwner = Access.hWndAccessApp

        .lStructSize = Len(Dialogue)

        .lpstrFilter = strFiltre

        .lpstrFile = Space(254)

        .nMaxFile = 255

        .lpstrFileTitle = Space(254)

        .nMaxFileTitle = 255

        .lpInitialFolder = InitialFolder

        .lpstrTitle = "Sauvegarde d'un fichier"

        .Flags = 6148

    End With

    RetVal = GetSaveFileName(Dialogue)

    If RetVal >= 1 Then

        SaveFile = Trim(Dialogue.lpstrFile)

        SaveFile = Replace(SaveFile, Chr(0), "")

    Else

        SaveFile = ""

        Exit Function

    End If

End Function

Avec Access 97

Version moins édulcorée, mais qui fonctionne (utilisée dans la base Cantor à VFE).
Trouvé sur le site de Jessy Sempere à la rubrique Fichiers/Sélectionner un fichier.

Ce code permet d'ouvrir une boîte de dialogue afin de pouvoir sélectionner un fichier.

Vous devez copier tout le code qui suit dans un module que vous pouvez appeler "modApiOpenFile".
 Ensuite vous n'avez plus qu'à appeler la fonction "OpenFile()".

L'argument "strInitialDir" représente le chemin par défaut lors de l'ouverture de la boîte de dialogue.

Fonction à insérer dans un module standard :

 '***************************Debut du code***************************

    Public Type OpenFileName

        lStructSize As Long

        hwndOwner As Long

        Instance As Long

        lpstrFilter As String

        lpstrCustomFilter As String

        nMaxCustomFilter As Long

        nFilterIndex As Long

        lpstrFile As String

        nMaxFile As Long

        lpstrFileTitle As String

        nMaxFileTitle As Long

        lpstrInitialDir As String

        lpstrTitle As String

        Flags As Long

        nFileOffset As Integer

        nFileExtension As Integer

        lpstrDefExt As String

        lCustData As Long

        lpfnHook As Long

        lpTemplateName As String

    End Type

    Public Declare Function GetOpenFileName Lib "comdlg32.dll" _

        Alias "GetOpenFileNameA" (pOpenfilename As OpenFileName) As Long

    Public Const OFN_AllowMultiSelect = &H200

    Public Const OFN_CreatePrompt = &H2000

    Public Const OFN_EnableHook = &H20

    Public Const OFN_EnableTemplate = &H40

    Public Const OFN_EnableTemplateHandle = &H80

    Public Const OFN_EXPLORER = &H80000

    Public Const OFN_ExtensionDifferent = &H400

    Public Const OFN_FileMustExist = &H1000

    Public Const OFN_HideReadOnly = &H4

    Public Const OFN_LongNames = &H200000

    Public Const OFN_NoChangeDir = &H8

    Public Const OFN_NoDeReferenceLinks = &H100000

    Public Const OFN_NoLongNames = &H40000

    Public Const OFN_NoNetWorkButton = &H20000

    Public Const OFN_NoReadOnlyReturn = &H8000

    Public Const OFN_NoTestFileCreate = &H10000

    Public Const OFN_NoValiDate = &H100

    Public Const OFN_OverWritePrompt = &H2

    Public Const OFN_PathMustExist = &H800

    Public Const OFN_ReadOnly = &H1

    Public Const OFN_ShareAware = &H4000

    Public Const OFN_ShareFallThrough = 2

    Public Const OFN_ShareNoWarn = 1

    Public Const OFN_ShareWarn = 0

    Public Const OFN_ShowHelp = &H10

    Global Dialogue As OpenFileName

    Public strFiltre As String

    Public strFile As String

    Public strNomFile As String

    Public RetVal As Long

    Public Function OpenFile(strInitialDir As String) As String

        OpenFile = ""

        strFiltre = "Fichiers Word" & Chr$(0) & "*.doc;*txt" & Chr$(0) & _

            "Fichiers Access" & Chr$(0) & "*.mdb" & Chr$(0) & _

            "Fichiers Excel" & Chr$(0) & "*.xls" & Chr$(0) & _

            "Tous les fichiers" & Chr$(0) & "*.*"

        With Dialogue

            .lStructSize = Len(Dialogue)

            .lpstrFilter = strFiltre

            .lpstrFile = Space(254)

            .nMaxFile = 255

            .lpstrFileTitle = Space(254)

            .nMaxFileTitle = 255

            .lpstrInitialDir = strInitialDir

            .lpstrTitle = "Recherche d'un fichier"

            .Flags = 6148 Or OFN_AllowMultiSelect Or OFN_LongNames Or OFN_EXPLORER

        End With

        RetVal = GetOpenFileName(Dialogue)

        If RetVal >= 1 Then

            OpenFile = Dialogue.lpstrFile

        Else

            OpenFile = ""

            Exit Function

        End If

    End Function

    '***************************Fin du code***************************

Copyright © PatCatNat's Page 10/11

http://www.patcatnats.fr/spip.php?article73


Boîte de sélection de chemins et ou de fichier(s)

Exemple d'appel de la fonction :
[BtnSelectChemin] est le bouton sur lequel l'évènement s'effectue.
[CheminBase] est le contrôle du formulaire qui doit contenir le chemin de la base de donnée sur laquelle travailler.

 Private Sub BtnSelectChemin_Click()

CheminBase = OpenFile("E:\Cantor")

End Sub

Post-scriptum :

– Toutes Api de Raymond sur son site
– Toutes Api de Jessy sur son site

